

## Aus der Excel-Trickkiste

# Fortgeschrittene Techniken für Projektleiter

## Teil 2: Portfolio-Auswahl mit dynamischen Diagrammen

Als Projektleiter haben Sie die Aufgabe, die Kosten und Fertigstellungsgrade mehrerer Projekte auszuwerten und in Diagrammform zu präsentieren. Ein zentrales Tabellenblatt soll dabei als Informationscockpit dienen. Welches Projekt dort angezeigt wird, lässt sich dabei aus einer Liste auswählen. Zusätzlich sollte auch die Übersicht über die Gesamtkosten aller Projekte möglich sein.

Auch für diese Aufgabenstellung aus dem Bereich Reporting und Präsentation lassen sich die im ersten Teil des Beitrags vorgestellten Listen, Dropdowns und Optionsfelder sinnvoll einsetzen. Dynamische Bereichsnamen und speziell die Matrixfunktion `BEREICH.VERSCHIEBEN()` sind für dynamische Diagramme die wichtigsten Komponenten.

Im vorgestellten Beispiel verwenden Sie Matrixfunktionen, um sowohl den Auswertungsbereich als auch das Chart für die grafische Analyse dynamisch zu gestalten – d.h. automatisch an den Datenbereich des ausgewählten Projekts anzupassen. Die vorgestellte Technik ist besonders hilfreich, wenn verschiedene Daten unterschiedlichen Umfangs grafisch umgesetzt werden sollen.



**Ignatz Schels**

selbstst. DV-Dozent und Journalist, konzipiert und leitet MS Project-Seminare, Projektberater und Autor

Kontakt: [info@schels.de](mailto:info@schels.de)

Mehr Informationen unter:  
[www.projektmagazin.de/autoren/](http://www.projektmagazin.de/autoren/)

## Vorbereitung für die Auswertung

In unserer Beispieldatei stehen die Projektdaten im Tabellenblatt "Projektportfolio" (Bild 1) – in der Praxis stammen sie jedoch meist aus unterschiedlichen Tabellenblättern oder Mappen. In diesem Fall sollten Sie die externen Daten per ODBC-Verknüpfung zusammenführen (siehe "Verteilte Excel-Daten mit ODBC zusammenführen", Schels, Projekt Magazin 13/2007).

IntelligenteFormel Elemente.xls [Kompatibilitätsmodus] - Microsoft Excel

DateiStartEinfügenSeitenlayoutFormelnDatenÜberprüfenAnsichtEntwicklertoolsAdd-InsPowerVivot

Einfügen

Calibri11A<sup>+</sup>

Standard

% 000

Bedingte Formatierung

Als Tabelle

Zellenformatvorlagen

Einfügen

Löschen

Format

Zellen

Sortieren und Filtern

Suchen und Auswählen

Bearbeiten

Zwischenabkl.

Schriftart

Ausrichtung

Zahl

L26

	A	B	C	D	E	F	G	H	I	J	K	L	M
1	Projekte	Projekt 1: Neugestaltung Kunsthalle Ost				Projekt 2: IT-Projekt Einführung CubeWare				Projekt 3: Bürokomplex Dietrichstraße			
2	Projekt 1: Neugestaltung Kunsthalle Ost												
3	Projekt 2: IT-Projekt Einführung CubeWare												
4	Projekt 3: Bürokomplex Dietrichstraße												
5		Januar	23.500		5%	Januar	45.000		10%	Januar	2.400		5%
6		Februar	21.000		12%	Februar	12.500		15%	Februar	31.000		20%
7		März	2.400		15%	März	17.800		20%	März	45.000		20%
8		April	31.000		15%	April	31.000		20%	April	21.000		35%
9		Mai	45.000		21%	Mai	30.000		25%	Mai	17.800		35%
10		Juni	30.000		24%	Juni	23.500		25%	Juni	31.000		40%
11		Juli	21.000		30%	Juli	21.000		25%	Juli			
12		August	12.500		40%	August	2.400		30%	August			
13		September	17.800		42%	September	31.000		50%	September			
14		Oktober				Oktober	21.000		50%	Oktober			
15		November				November	21.000		50%	November			
16		Dezember				Dezember				Dezember			
17													
18													
19													
20													

Projektportfolio

Bereit

100 %

Bild 1: Projektportfolio mit drei Projekten.

## Alle Projekte zusammenfassen

Um in der Auswertung des Portfolios auch die Gesamtkosten aller Projekte zu berücksichtigen, werden diese in den Spalten neben den Projektdaten berechnet. Für die Fertigungsgrade erstellen Sie eine Mittelwertberechnung.

Im Beispiel legen Sie dazu zunächst im Tabellenblatt "Projektportfolio" eine Monatsreihe im Bereich \$O\$4:\$Q\$15 an. Anschließend füllen Sie die Zellen wie folgt aus:

A5: Alle Projekte

P3: Ist-Kosten

Q3: Fertigstellung (Ø)

P4: =SUMME(D4;H4;L4)

Q4: =WENN(P4>0;MITTELWERT(E4;I4;M4);0)

Zum Abschluss kopieren Sie die Formeln für Summe und Mittelwert mit dem Füllkästchen am Zellzeiger nach unten bis zum Ende des Bereichs.

## Bereichsnamen zuweisen

Um die Projektdaten in die Auswertung zu übernehmen, werden ihnen mit Hilfe des Namens-Managers (STRG+F3) Bereichsnamen zugewiesen. Diese lassen sich, wie im ersten Teil des Beitrags gezeigt, mit BEREICH.VERSCHIEBEN() berechnen oder können – wenn ein einheitlicher Auswertungszeitraum von zwölf Monaten wie hier im Beispiel vorliegt – mit absolutem Zellbezug zugewiesen werden:

Bereichsname	Bezieht sich auf
Projekt1	=Projektportfolio!\$C\$4:\$E\$15
Projekt2	=Projektportfolio!\$G\$4:\$I\$15
Projekt3	=Projektportfolio!\$K\$4:\$M\$15
AlleProjekte	=Projektportfolio!\$O\$4:\$Q\$15
Projekte	=Projektportfolio!\$A\$2:\$A\$5

## Projekttabellen in das Informationscockpit übertragen

Im Beispiel dient das Tabellenblatt "ProjektINFO" als Informationscockpit, in dem die ausgewählten Projektdaten angezeigt werden.

## Projektauswahl durch Kombinations- und Drehfeld

Der Kopfbereich des Informationscockpits erhält ein Kombinationsfeld und ein Drehfeld zur Projektauswahl (Bild 2). Beide Felder haben die gleiche Ausgabeverknüpfung (\$B\$5). Das Kombinationsfeld erhält als Eingabebereich den Bereich "Projekte", der Maximalwert des Drehfelds wird manuell auf die Anzahl der Projekte beschränkt. Den Maximalwert dynamisch anzupassen, wäre nur mit Einsatz von VBA-Makros möglich, da die Eigenschaftenbox keine Formeln akzeptiert.

In Zeile 7 werden die Spaltennamen für die Auswertung der Projektdaten eingetragen (Bild 2).



Bild 2: Projektauswahl durch Kombinations- und Drehfeld.

## Projektdaten mit Matrixformel übertragen

Datenbereiche lassen sich mit Hilfe der Matrixformel INDEX() und unter Verwendung eines Listenelements und dessen Ausgabeverknüpfung in andere Formularelemente transferieren, wie die Beispiele in Teil 1 gezeigt haben. Das funktioniert nicht nur mit einzelnen Datenreihen, sondern auch mit ganzen Tabellenbereichen.

Die Aufgabe besteht darin, die Daten des im Kombinationsfeld ausgewählten Projekts in die Tabelle ProjektINFO zu übertragen. Legen Sie dazu einen Bereichsnamen "Projekt" an, der die Funktion WAHL() verwendet. Startparameter in der Funktion ist die Ausgabeverknüpfung des Kombinationsfelds, als weitere Argumente werden die Namen der einzelnen Projekte eingetragen:

Bereichsname	Bezieht sich auf
Projekt	=WAHL(Projekt!INFO!\$A\$5;Projekt1;Projekt2;Projekt3;AlleProjekte)

Markieren Sie den Bereich, in dem die Daten abzubilden sind (im Beispiel A8:C19), tippen Sie die nachfolgend genannte Formel ein und drücken Sie STRG+EINGABE, um die Formel auf den markierten Bereich zu übertragen:

A8:

=INDEX(Projekt;ZEILE()-7;SPALTE())

Die Funktion INDEX() indiziert den Bereich mit Zeile und Spalte. Den passenden Index liefert dabei die Zeilen- und die Spaltennummer der Zelle, in der die Formel steht. Da der Zielbereich in Zeile 8 beginnt, muss der Zeilenindex in der Formel um den Wert 7 verringert werden ("ZEILE()-7").

Damit der Bereich nur die bereits erfassten Daten der Projekte abbildet, erweitern Sie die Formel mit einer WENN()-Funktion, die nur Werte größer 0 zulässt:

A8: =WENN(INDEX(Projekt;ZEILE()-7;SPALTE())>0;INDEX(Projekt;ZEILE()-7;SPALTE());"")

Schreiben Sie in der nächsten Spalte neben dem Zielbereich die Formel zur Kumulierung der Monatswerte und kopieren Sie die Formeln der einzelnen Hilfsreihen mit dem Füllkästchen am Zellzeiger nach unten bis zum Ende des Bereichs:

D8: =WENN(B8<>"";SUMME(\$B\$8:B8);0)

Die Kostensumme berechnen Sie in einer Zelle im Kopfbereich:

G3: =SUMME(\$B\$8:\$B\$19)

Monat	Kosten	FstG	Kosten kumuliert
Januar	23.500	5%	23.500
Februar	21.000	12%	44.500
März	2.400	15%	46.900
April	31.000	15%	77.900
Mai	45.000	21%	122.900
Juni	30.000	24%	152.900
Juli	21.000	30%	173.900
August	12.500	40%	186.400
September	17.800	42%	204.200
Oktober			
November			
Dezember			

Bild 3: Die Matrixfunktion INDEX() transferiert die Daten des ausgewählten Projekts in den Zielbereich.

## Dynamisches Säulen-/Liniendiagramm

### Ein Diagramm mit absoluten Bezügen erstellen

Um die ausgewählten Projektdaten anschaulich darzustellen, stellen Sie kumulierte Kosten und Fertigstellungsgrad in einem Diagramm einander gegenüber. Als Diagrammtyp eignet sich dafür am besten eine Kombination aus Säulen und Linien. Erstellen Sie zunächst ein Säulendiagramm mit absoluten Zellbezügen:

Markieren Sie dazu zunächst den Datenbereich für die Monate (im Beispiel: \$A\$7:\$A\$13, siehe Bild 4), halten Sie die Strg-Taste gedrückt und markieren Sie anschließend den Bereich für die kumulierten Kosten (im Beispiel: \$D\$7:\$D\$13). Erzeugen Sie mit *Einfügen / Diagramme / Säule / Gruppierte Säulen* ein Säulendiagramm im Tabellenblatt.

Markieren Sie jetzt den Bereich für die zweite Datenreihe mit dem Fertigstellungsgrad (im Beispiel: \$C\$7:\$C\$13) und kopieren Sie die markierten Daten mit STRG+C in die Zwischenablage. Markieren Sie das Diagrammobjekt und drücken Sie die EINGABE-Taste, um die neue Reihe einzufügen.

Damit der Fertigstellungsgrad auf einer separaten Achse aufgetragen wird, weisen Sie der zweiten Reihe einen neuen Diagrammtyp zu und aktivieren für diesen die Sekundärachse (Bild 4).

Die Befehle dafür lauten für Excel bis Version 2003:

*Diagramm / Diagrammtyp Linie*

*Diagramm / Diagrammoptionen / Achsen / Sekundärachse / Größenachse*

Für Excel 2007 und 2010:

*Diagrammtools / Entwurf / Typ / Diagrammtyp ändern / Linie*

*Diagrammtools / Layout / Aktuelle Auswahl / Auswahl formatieren / Reihenoptionen / Datenreihe zeichnen auf Sekundärachse.*




Bild 4: Diagramm mit Primär- und Sekundärachse.

## Die Funktion DATENREIHE()

Der Grundstein aller Diagramme ist die Funktion DATENREIHE(). Excel definiert jede Reihe, egal ob Säule, Balken, Linie oder Kreis mit Hilfe dieser Funktion aus den Daten im Tabellenblatt. Ein Klick auf die erste Datenreihe blendet die Funktion in der Bearbeitungsleiste ein.

`=DATENREIHE(ProjektINFO!$C$7;ProjektINFO!$A$8:$A$13;ProjektINFO!$C$8:$C$13;2)`

Das erste Argument verknüpft bei Diagrammen mit nur einer Reihe eine Zelle mit dem Diagrammtitel. Enthält ein Diagramm mehr als eine Reihe, bildet diese Verknüpfung den Legendeneintrag.

Das zweite Argument liefert die Beschriftung der horizontalen Achse (X-Achse), das dritte definiert die Daten für die Größenachse, d.h. für die Länge der Balken oder die Position der Datenpunkte in Linien- und Punktereihen. Das vierte Argument wird für die Position der Reihe im Diagramm benötigt, 1 bezeichnet die erste Reihe von links, 2 die zweite usw.

## Absolute durch dynamische Bezüge ersetzen

Damit das Diagramm nur so viele Datenpunkte anzeigt, wie Werte zu den einzelnen Monaten im Tabellenblatt stehen, legen Sie für jede Diagrammreihe einen dynamischen Bereichsnamen an. In diesem Bereichsnamen stellt die Funktion BEREICH.VERSCHIEBEN() in Kombination mit ANZAHL() sicher, dass der Name nur genau die Werte einschließt, die in den Spalten C und D zu finden sind. Die Funktion ANZAHL() berechnet dazu die Anzahl der Spalteneinträge und liefert den entsprechenden Wert an die Matrixfunktion BEREICH.VERSCHIEBEN().



Legen Sie die Bereichsnamen für das Diagramm "lokal" an, so dass Sie nur für die Tabelle ProjektINFO gelten.

Lokale Bereichsnamen sind nur für das jeweilige Tabellenblatt verfügbar und werden nur für dieses angezeigt oder angeboten. Wird der Name für den Rubriken- oder Datenbereich eines Diagramms verwendet, bleibt die Verknüpfung auf das Tabellenblatt erhalten. Bei globalen Namen schaltet Excel die Verknüpfung automatisch auf die gesamte Mappe um.

Um den Bereichsnamen lokal anzulegen, schreiben Sie in Excel 2003 vor den Bereichsnamen den Namen des Blatts: "ProjektINFO!Kosten". In Excel 2007 und 2010 schalten Sie nach Aufruf des Namens-Managers unter *Neu / Bereich* auf das Tabellenblatt um, in dem sich die Daten befinden.

Bereichsname	Bezieht sich auf
Rubrik	=BEREICH.VERSCHIEBEN(PprojektINFO!\$A\$8;0;0;ANZAHL(PprojektINFO!\$B:\$B);1)
Kosten	=BEREICH.VERSCHIEBEN(PprojektINFO!\$B\$8;0;0;ANZAHL(PprojektINFO!\$B:\$B);1)
FstG	=BEREICH.VERSCHIEBEN(PprojektINFO!\$C\$8;0;0;ANZAHL(PprojektINFO!\$C:\$C);1)
KostenKum	=BEREICH.VERSCHIEBEN(PprojektINFO!\$D\$8;0;0;ANZAHL(PprojektINFO!\$B:\$B);1)

Ändern Sie anschließend in der Bearbeitungsleiste die Formel für die einzelnen Datenreihen ab, indem Sie dort statt der absoluten Bezüge die soeben definierten Bereichsnamen verwenden. Markieren Sie dazu die erste Diagrammreihe und geben Sie in der Bearbeitungsleiste ein:

=DATENREIHE(PprojektINFO!\$B\$7;PprojektINFO!Rubrik;PprojektINFO!KostenKum;1)

Die zweite Datenreihe erhält ebenfalls die berechneten Bereichsnamen für die Rubriken- und Größenachse an Stelle der absoluten Bezüge:

=DATENREIHE(PprojektINFO!\$C\$7;PprojektINFO!Rubrik;PprojektINFO!FstG;2)

## Diagrammtitel dynamisch erzeugen

Da das Diagramm unterschiedliche, von der Auswahl im Listenfeld abhängige Daten abbildet, muss auch die Überschrift angepasst werden. Konstruieren Sie für den Diagrammtitel den Namen des gewählten Projekts in einer für den Benutzer unsichtbaren Zelle, am besten hinter dem Listenfeld, nutzen Sie die Funktion INDEX() und die Ausgabeverknüpfung des Listenfeldes:

A3: =INDEX(Pprojekte;\$A\$5;1)

Verknüpfen Sie den Titelbereich des Diagramms anschließend mit diesem berechneten Projektnamen.

Für Excel bis Version 2003 lautet der Befehl dafür: *Diagramm / Diagrammoptionen / Titel*. Schreiben Sie in das Feld "Diagrammtitel" einen beliebigen Text und bestätigen Sie mit OK. Markieren Sie dann das Titелеlement, schreiben Sie ein "="-Zeichen in die Bearbeitungszeile und klicken Sie anschließend auf die Zelle A3. Nach Bestätigung mit EINGABE ist der Titel mit der Formel verknüpft.

In Excel 2007 und 2010 zeichnen Sie mit *Einfügen / Text / Textfeld* ein Textfeld in das markierte Diagrammobjekt. Markieren Sie das Textfeld am Rand, klicken Sie in die Bearbeitungsleiste und schreiben Sie die Verknüpfung:

=PprojektINFO!\$A\$3




Bild 5: Diagramm mit dynamischen Bereichen.

## Dynamisches Wasserfalldiagramm

Das zweite Diagramm im Tabellenblatt ProjektINFO informiert den Anwender in Form eines Wasserfalls über die monatliche Veränderung der Projektkosten. Steigende Kosten werden als rote, fallende als grüne Balken dargestellt. Damit alle Werte positiv auf einer einheitlichen X-Achse abgebildet werden, sind einige Zwischenberechnungen erforderlich (siehe Bild 6, ab Spalte O).



Schreiben Sie keine weiteren Werte oder Berechnungen in die Spalten B und C, da diese für die dynamischen Bereichsnamen ausgewertet werden.

### Zwischenberechnungen

Berechnen Sie ab Spalte O die Veränderungen, unterteilt in steigende und fallende Kosten (Bild 6). Die Zeile nach der Überschrift bleibt leer, damit die Veränderung als Differenz zwischen Monat und Vormonat auch bereits für den ersten Monat korrekt berechnet wird.

O24: Monat

P24: Kosten

Q24: Veränderung

R24: steigend

S24: fallend

T24: Titel

Die erste Hilfsreihe berechnet die Namen der Monate, die in der Kostenaufstellung angezeigt werden, die zweite Hilfsreihe zeigt die angefallenen Kosten.

O26: =WENN(A8<>"";A8;"")

P26: =WENN(B8<>"";B8;"")

Bild 6: Zwischenberechnungen für das Wasserfalldiagramm.

In der dritten bis fünften Hilfsreihe werden die Veränderungen berechnet. Damit das Diagramm auch die fallenden Werte als positive Werte anzeigt, setzen Sie für diese ein Minuszeichen vor den Zellbezug.

Q26: =WENN(P26<>"";P26-P25;0)

R26: =WENN(Q26>0;Q26;0)

S26: =WENN(Q26<0;-Q26;0)

Mit dem Füllkästchen am Zellzeiger werden die Formeln der einzelnen Hilfsreihen nach unten kopiert.

### Diagrammtitel

Der Titel für das Diagramm sollte zweizeilig sein. In der ersten Zeile steht die jeweilige Projektbezeichnung, in der zweiten ein Hinweis auf die Art der Darstellung. Um den Titel dynamisch zu erzeugen, verknüpfen Sie in Zelle T25 den Projektnamen aus der Zelle A3 mit einem Zeilenumbruch und dem Text für die zweite Zeile:

T25: =A3&ZEICHEN(10)&"Projektkosten – monatliche Veränderung"



Die Funktion ZEICHEN() liefert das Zeichen, dessen ASCII-Wert in der Funktionsklammer angegeben wird. Der ASCII-Wert 10 ist ein nicht druckbares Steuerzeichen für Zeilenumbruch (eigentlich 10 + 13 für carriage return und line feed, aber ZEICHEN(10) genügt, um im Text einen Zeilenumbruch herbeizuführen).

	T25								
	O	P	Q	R	S	T	U	V	W
23									
24	Monat	Kosten	Veränderung	steigend	fallend	Titel			
25						Projekt 1: Neugestaltung Kunsthalle Ost	Projektk		
26	Januar	23.500	23.500	23.500					
27	Februar	21.000	-2.500		2.500				
28	März	2.400	-18.600		18.600				
29	April	31.000	28.600	28.600					
30	Mai	45.000	14.000	14.000					
31	Juni	30.000	-15.000		15.000				
32	Juli	21.000	-9.000		9.000				
33	August	12.500	-8.500		8.500				
34	September	17.800	5.300	5.300					
35	Oktober								
36	November								
37	Dezember								



## Datenreihen für steigende und fallende Kosten

Legen Sie für das neue Diagramm als erstes wieder ein Diagramm mit festen Zellbezügen an und passen Sie es anschließend an die dynamisch kalkulierten Reihen an. Markieren Sie dazu zunächst die Monate, für die Kosten angefallen sind, dann die zugehörigen Kosten (im Beispiel O26:O34 und P26:P34, siehe Bild 6) und erstellen Sie ein Säulendiagramm vom Typ *Gestapelte Säulen* im 2D-Format.

Damit das Diagramm die Monatskosten dynamisch anzeigt, weisen Sie der ersten Datenreihe die für das Säulen-/Liniendiagramm bereits definierten Bereichsnamen "Rubrik" und "Kosten" zu.

Die Formel der ersten Datenreihe lautet dann:

=DATENREIHE(;ProjektINFO!Rubrik;ProjektINFO!Kosten;1)

Legen Sie zusätzlich zwei neue lokale Bereichsnamen für die steigenden und fallenden Werte an:

Bereichsname	Bezieht sich auf
ProjektINFO!WF_fallend	=BEREICH.VERSCHIEBEN(ProjektINFO!\$S\$26;0;0;ZEILEN(ProjektINFO!Rubrik);1)
ProjektINFO!WF_steigend	=BEREICH.VERSCHIEBEN(ProjektINFO!\$R\$26;0;0;ZEILEN(ProjektINFO!Rubrik);1)

Um diese beiden Bereichsnamen als Datenreihen in das Diagramm zu bekommen, erstellen Sie in dem soeben erzeugten Säulendiagramm zunächst zwei weitere Datenreihen mit festem Zellbezug. Ziehen Sie dazu die Markierung über die Zellen mit den Werten "Veränderung steigend" (R26:R34), kopieren Sie diese mit Strg+C in die Zwischenablage, markieren Sie das Diagramm und fügen Sie die neue Reihe mit EINGABE ein. Verfahren Sie dann analog mit den Werten für "Veränderung fallend".

Passen Sie anschließend die Datenreihe-Funktionen dieser beiden Diagrammreihen an und weisen Sie ihnen jeweils im ersten Argument die Rubrik und im zweiten Argument die steigenden bzw. fallenden Werte zu. Der Legendentext wird aus der Überschrift der Hilfsreihen gebildet. Die Formeln der zweiten und dritten Datenreihe lauten dann:

=DATENREIHE(ProjektINFO!\$R\$24;ProjektINFO!Rubrik;ProjektINFO!WF\_steigend;2)

=DATENREIHE(ProjektINFO!\$S\$24;ProjektINFO!Rubrik;ProjektINFO!WF\_fallend;3)

Jetzt wird die erste Reihe ausgeblendet, dazu schalten Sie die Füllung und die Rahmenformatierung aus. Fügen Sie noch die Datenbeschriftungen für die beiden übrigen Reihen hinzu und entfernen Sie überflüssige Elemente wie Gitternetzlinien und Rahmen. Die Achse wird wie im ersten Diagramm auf die Skalierungseinheit *Tausende* gesetzt. Der zweizeilige Titel entsteht aus der Verknüpfung auf die zuvor erstellte Formel in Zelle T25.




Bild 7: Wasserfalldiagramm mit steigenden und fallenden Kosten.

## Diagrammvariante: Pfeilsymbole

Eine Alternative zu bekannten Diagrammtypen wie Säulen oder Balken lässt sich mit einem kleinen Trick herstellen: Kopieren Sie ein gezeichnetes Element oder eine eingefügte Grafik einfach auf die Diagrammreihe. Hier zum Beispiel Pfeilsymbole für steigende und fallende Kosten:

Zeichnen Sie mit Hilfe der Zeichenwerkzeuge einen grünen, nach oben zeigenden Pfeil und einen roten, nach unten zeigenden Pfeil. Die Zeichenwerkzeuge finden Sie in Excel bis Version 2003 in der Symbolleiste *Zeichnen* und in Excel 2007 / 2010 unter *Einfügen / Illustrationen / Formen*.

Markieren Sie den ersten Pfeil und kopieren Sie ihn mit Strg+C in die Zwischenablage. Markieren Sie anschließend die Datenreihe im Diagramm und setzen Sie mit Strg+V den Pfeil auf die einzelnen Datenpunkte. Setzen Sie so auch den zweiten Pfeil auf die nächste Datenreihe.




Bild 8: Pfeilsymbole für steigende und fallende Kosten.

### Hat Ihnen dieser Artikel gefallen?

Bewerten Sie ihn im Projekt Magazin online und teilen Sie so Ihre Meinung anderen Lesern mit. Wählen Sie dazu den Artikel im Internet unter [www.projektmagazin.de/ausgaben/2010](http://www.projektmagazin.de/ausgaben/2010) oder klicken Sie [hier](#), um direkt zum Artikel zu gelangen.